
Supplementary Material S3 

SPSS Analysis Syntax 

Software: IBM SPSS Statistics, version 26. 

 

The following syntax reproduces the principal and sensitivity analyses reported in the manuscript. Analyses 

were conducted on the trial-level dataset containing 440 observations (22 participants × 20 experimental 

questions). 

 

Variable coding used in the analysis dataset: 

 

participant_id      Numerical participant identifier 

condition           0 = Baseline block; 1 = Auditory-cue block 

period              1 = First experimental period; 2 = Second experimental period 

sequence            0 = AB; 1 = BA 

item_id             Experimental question identifier (1–20) 

question_number     Question position within block (1–10) 

knowledge_context   0 = International/general; 1 = Ecuador-related 

thematic_pair       Corresponding thematic pair (1–10) 

correct             0 = Incorrect/response expiration; 1 = Correct 

timeout             0 = Response submitted; 1 = Response expiration 

response_time_s     Response time in seconds for submitted responses 

 

The technical variable "timeout" corresponds to the outcome referred to as "response expiration" throughout the 

manuscript. 

 

 

*********************************************************************** 

1. MAIN ALL-TRIAL ACCURACY MODEL 

***********************************************************************. 

 

GENLINMIXED 

 /DATA_STRUCTURE SUBJECTS=participant_id*item_id 

 /FIELDS TARGET=correct 

 /TARGET_OPTIONS REFERENCE=0 DISTRIBUTION=BINOMIAL LINK=LOGIT 

 /FIXED EFFECTS=condition knowledge_context period 

                condition*knowledge_context 

        USE_INTERCEPT=TRUE 

 /RANDOM USE_INTERCEPT=TRUE 

        SUBJECTS=participant_id 

        COVARIANCE_TYPE=VARIANCE_COMPONENTS 

 /RANDOM USE_INTERCEPT=TRUE 

        SUBJECTS=item_id 

        COVARIANCE_TYPE=VARIANCE_COMPONENTS 

 /BUILD_OPTIONS TARGET_CATEGORY_ORDER=ASCENDING 

        INPUTS_CATEGORY_ORDER=ASCENDING 



        MAX_ITERATIONS=100 

        CONFIDENCE_LEVEL=95 

        DF_METHOD=SATTERTHWAITE 

        COVB=MODEL 

 /EMMEANS TABLES=condition 

 /EMMEANS TABLES=condition*knowledge_context 

        COMPARE=condition CONTRAST=PAIRWISE 

 /EMMEANS_OPTIONS SCALE=ORIGINAL PADJUST=SEQBONFERRONI. 

 

* Odds ratios and 95% confidence intervals were obtained by exponentiating 

* the corresponding marginal logit contrast and its confidence limits. 

 

 

*********************************************************************** 

2. REDUCED ACCURACY MODEL WITHOUT THE INTERACTION 

***********************************************************************. 

 

GENLINMIXED 

 /DATA_STRUCTURE SUBJECTS=participant_id*item_id 

 /FIELDS TARGET=correct 

 /TARGET_OPTIONS REFERENCE=0 DISTRIBUTION=BINOMIAL LINK=LOGIT 

 /FIXED EFFECTS=condition knowledge_context period 

        USE_INTERCEPT=TRUE 

 /RANDOM USE_INTERCEPT=TRUE 

        SUBJECTS=participant_id 

        COVARIANCE_TYPE=VARIANCE_COMPONENTS 

 /RANDOM USE_INTERCEPT=TRUE 

        SUBJECTS=item_id 

        COVARIANCE_TYPE=VARIANCE_COMPONENTS 

 /BUILD_OPTIONS TARGET_CATEGORY_ORDER=ASCENDING 

        INPUTS_CATEGORY_ORDER=ASCENDING 

        MAX_ITERATIONS=100 

        CONFIDENCE_LEVEL=95 

        DF_METHOD=SATTERTHWAITE 

        COVB=MODEL 

 /EMMEANS TABLES=condition COMPARE=condition CONTRAST=PAIRWISE 

 /EMMEANS_OPTIONS SCALE=ORIGINAL PADJUST=LSD. 

 

 

*********************************************************************** 

3. MAIN RESPONSE-EXPIRATION MODEL 

***********************************************************************. 

 

GENLINMIXED 

 /DATA_STRUCTURE SUBJECTS=participant_id*item_id 

 /FIELDS TARGET=timeout 

 /TARGET_OPTIONS REFERENCE=0 DISTRIBUTION=BINOMIAL LINK=LOGIT 

 /FIXED EFFECTS=condition knowledge_context period 



                condition*knowledge_context 

        USE_INTERCEPT=TRUE 

 /RANDOM USE_INTERCEPT=TRUE 

        SUBJECTS=participant_id 

        COVARIANCE_TYPE=VARIANCE_COMPONENTS 

 /RANDOM USE_INTERCEPT=TRUE 

        SUBJECTS=item_id 

        COVARIANCE_TYPE=VARIANCE_COMPONENTS 

 /BUILD_OPTIONS TARGET_CATEGORY_ORDER=ASCENDING 

        INPUTS_CATEGORY_ORDER=ASCENDING 

        MAX_ITERATIONS=100 

        CONFIDENCE_LEVEL=95 

        DF_METHOD=SATTERTHWAITE 

        COVB=MODEL 

 /EMMEANS TABLES=condition 

 /EMMEANS TABLES=condition*knowledge_context 

        COMPARE=condition CONTRAST=PAIRWISE 

 /EMMEANS_OPTIONS SCALE=ORIGINAL PADJUST=SEQBONFERRONI. 

 

* Odds ratios and 95% confidence intervals were obtained by exponentiating 

* the corresponding marginal logit contrast and its confidence limits. 

 

 

*********************************************************************** 

4. PARTICIPANT-LEVEL PAIRED RESPONSE-EXPIRATION SUMMARY 

***********************************************************************. 

 

DATASET NAME TrialData. 

DATASET COPY ExpirationCounts. 

DATASET ACTIVATE ExpirationCounts. 

 

AGGREGATE 

 /OUTFILE=* MODE=REPLACE 

 /BREAK=participant_id condition 

 /exp_count=SUM(timeout). 

 

SORT CASES BY participant_id condition. 

 

CASESTOVARS 

 /ID=participant_id 

 /INDEX=condition 

 /GROUPBY=VARIABLE 

 /SEPARATOR="_". 

 

* Assumes condition 0 = baseline and condition 1 = auditory-cue block. 

COMPUTE expiration_difference=exp_count_0-exp_count_1. 

 

* Positive = fewer expirations in auditory-cue block. 



* Zero     = equal number of expirations. 

* Negative = more expirations in auditory-cue block. 

 

COMPUTE expiration_pattern=0. 

IF (expiration_difference>0) expiration_pattern=1. 

IF (expiration_difference<0) expiration_pattern=-1. 

 

VALUE LABELS expiration_pattern 

 -1 'More expirations in auditory-cue block' 

  0 'Equal number of expirations' 

  1 'Fewer expirations in auditory-cue block'. 

 

FREQUENCIES VARIABLES=expiration_pattern. 

 

DESCRIPTIVES VARIABLES=exp_count_0 exp_count_1 expiration_difference 

 /STATISTICS=MEAN STDDEV MIN MAX. 

 

T-TEST PAIRS=exp_count_0 WITH exp_count_1 (PAIRED) 

 /CRITERIA=CI(.95). 

 

DATASET ACTIVATE TrialData. 

 

 

*********************************************************************** 

5. RESPONSE-TIME TRANSFORMATION AND MAIN MODEL 

***********************************************************************. 

 

COMPUTE ln_response_time=$SYSMIS. 

IF (timeout=0 AND response_time_s>0) 

    ln_response_time=LN(response_time_s). 

EXECUTE. 

 

GENLINMIXED 

 /DATA_STRUCTURE SUBJECTS=participant_id*item_id 

 /FIELDS TARGET=ln_response_time 

 /TARGET_OPTIONS DISTRIBUTION=NORMAL LINK=IDENTITY 

 /FIXED EFFECTS=condition knowledge_context period 

                condition*knowledge_context 

        USE_INTERCEPT=TRUE 

 /RANDOM USE_INTERCEPT=TRUE 

        SUBJECTS=participant_id 

        COVARIANCE_TYPE=VARIANCE_COMPONENTS 

 /RANDOM USE_INTERCEPT=TRUE 

        SUBJECTS=item_id 

        COVARIANCE_TYPE=VARIANCE_COMPONENTS 

 /BUILD_OPTIONS INPUTS_CATEGORY_ORDER=ASCENDING 

        MAX_ITERATIONS=100 

        CONFIDENCE_LEVEL=95 



        DF_METHOD=SATTERTHWAITE 

        COVB=MODEL 

 /EMMEANS TABLES=condition 

 /EMMEANS TABLES=condition*knowledge_context 

        COMPARE=condition CONTRAST=PAIRWISE 

 /EMMEANS_OPTIONS SCALE=ORIGINAL PADJUST=SEQBONFERRONI. 

 

* Only trials with submitted responses contribute to this model because 

* ln_response_time is missing for response-expiration trials. 

* 

* Geometric means were obtained by exponentiating estimated marginal means 

* from the log scale. The geometric mean ratio (GMR) and its 95% confidence 

* interval were obtained by exponentiating the corresponding condition 

* contrast and its confidence limits. 

 

 

*********************************************************************** 

6. ACCURACY AMONG SUBMITTED RESPONSES 

***********************************************************************. 

 

TEMPORARY. 

SELECT IF (timeout=0). 

 

GENLINMIXED 

 /DATA_STRUCTURE SUBJECTS=participant_id*item_id 

 /FIELDS TARGET=correct 

 /TARGET_OPTIONS REFERENCE=0 DISTRIBUTION=BINOMIAL LINK=LOGIT 

 /FIXED EFFECTS=condition knowledge_context period 

                condition*knowledge_context 

        USE_INTERCEPT=TRUE 

 /RANDOM USE_INTERCEPT=TRUE 

        SUBJECTS=participant_id 

        COVARIANCE_TYPE=VARIANCE_COMPONENTS 

 /RANDOM USE_INTERCEPT=TRUE 

        SUBJECTS=item_id 

        COVARIANCE_TYPE=VARIANCE_COMPONENTS 

 /BUILD_OPTIONS TARGET_CATEGORY_ORDER=ASCENDING 

        INPUTS_CATEGORY_ORDER=ASCENDING 

        MAX_ITERATIONS=100 

        CONFIDENCE_LEVEL=95 

        DF_METHOD=SATTERTHWAITE 

        COVB=MODEL 

 /EMMEANS TABLES=condition 

 /EMMEANS TABLES=condition*knowledge_context 

        COMPARE=condition CONTRAST=PAIRWISE 

 /EMMEANS_OPTIONS SCALE=ORIGINAL PADJUST=SEQBONFERRONI. 

 

 



*********************************************************************** 

7. ORIGINAL 15-S DEADLINE SENSITIVITY OUTCOME 

***********************************************************************. 

 

COMPUTE original15_failure=0. 

IF (timeout=1) original15_failure=1. 

IF (timeout=0 AND response_time_s>15) original15_failure=1. 

EXECUTE. 

 

GENLINMIXED 

 /DATA_STRUCTURE SUBJECTS=participant_id*item_id 

 /FIELDS TARGET=original15_failure 

 /TARGET_OPTIONS REFERENCE=0 DISTRIBUTION=BINOMIAL LINK=LOGIT 

 /FIXED EFFECTS=condition knowledge_context period 

                condition*knowledge_context 

        USE_INTERCEPT=TRUE 

 /RANDOM USE_INTERCEPT=TRUE 

        SUBJECTS=participant_id 

        COVARIANCE_TYPE=VARIANCE_COMPONENTS 

 /RANDOM USE_INTERCEPT=TRUE 

        SUBJECTS=item_id 

        COVARIANCE_TYPE=VARIANCE_COMPONENTS 

 /BUILD_OPTIONS TARGET_CATEGORY_ORDER=ASCENDING 

        INPUTS_CATEGORY_ORDER=ASCENDING 

        MAX_ITERATIONS=100 

        CONFIDENCE_LEVEL=95 

        DF_METHOD=SATTERTHWAITE 

        COVB=MODEL 

 /EMMEANS TABLES=condition COMPARE=condition CONTRAST=PAIRWISE 

 /EMMEANS_OPTIONS SCALE=ORIGINAL PADJUST=LSD. 

 

 

*********************************************************************** 

8. RESPONSE TIME EXCLUDING TIMER-RESET TRIALS 

***********************************************************************. 

 

* Replace timer_reset with the exact timer-reset indicator variable name 

* used in the analysis dataset if different. 

 

TEMPORARY. 

SELECT IF (timeout=0 AND timer_reset=0). 

 

GENLINMIXED 

 /DATA_STRUCTURE SUBJECTS=participant_id*item_id 

 /FIELDS TARGET=ln_response_time 

 /TARGET_OPTIONS DISTRIBUTION=NORMAL LINK=IDENTITY 

 /FIXED EFFECTS=condition knowledge_context period 

                condition*knowledge_context 



        USE_INTERCEPT=TRUE 

 /RANDOM USE_INTERCEPT=TRUE 

        SUBJECTS=participant_id 

        COVARIANCE_TYPE=VARIANCE_COMPONENTS 

 /RANDOM USE_INTERCEPT=TRUE 

        SUBJECTS=item_id 

        COVARIANCE_TYPE=VARIANCE_COMPONENTS 

 /BUILD_OPTIONS INPUTS_CATEGORY_ORDER=ASCENDING 

        MAX_ITERATIONS=100 

        CONFIDENCE_LEVEL=95 

        DF_METHOD=SATTERTHWAITE 

        COVB=MODEL 

 /EMMEANS TABLES=condition 

 /EMMEANS_OPTIONS SCALE=ORIGINAL PADJUST=LSD. 

 

 

*********************************************************************** 

9. PRESENTATION-SEQUENCE SENSITIVITY ANALYSES 

***********************************************************************. 

 

* Accuracy. 

 

GENLINMIXED 

 /DATA_STRUCTURE SUBJECTS=participant_id*item_id 

 /FIELDS TARGET=correct 

 /TARGET_OPTIONS REFERENCE=0 DISTRIBUTION=BINOMIAL LINK=LOGIT 

 /FIXED EFFECTS=condition knowledge_context period 

                condition*knowledge_context sequence 

        USE_INTERCEPT=TRUE 

 /RANDOM USE_INTERCEPT=TRUE SUBJECTS=participant_id 

        COVARIANCE_TYPE=VARIANCE_COMPONENTS 

 /RANDOM USE_INTERCEPT=TRUE SUBJECTS=item_id 

        COVARIANCE_TYPE=VARIANCE_COMPONENTS 

 /BUILD_OPTIONS CONFIDENCE_LEVEL=95 

        DF_METHOD=SATTERTHWAITE COVB=MODEL. 

 

* Response expiration. 

 

GENLINMIXED 

 /DATA_STRUCTURE SUBJECTS=participant_id*item_id 

 /FIELDS TARGET=timeout 

 /TARGET_OPTIONS REFERENCE=0 DISTRIBUTION=BINOMIAL LINK=LOGIT 

 /FIXED EFFECTS=condition knowledge_context period 

                condition*knowledge_context sequence 

        USE_INTERCEPT=TRUE 

 /RANDOM USE_INTERCEPT=TRUE SUBJECTS=participant_id 

        COVARIANCE_TYPE=VARIANCE_COMPONENTS 

 /RANDOM USE_INTERCEPT=TRUE SUBJECTS=item_id 



        COVARIANCE_TYPE=VARIANCE_COMPONENTS 

 /BUILD_OPTIONS CONFIDENCE_LEVEL=95 

        DF_METHOD=SATTERTHWAITE COVB=MODEL. 

 

* Response time among submitted responses. 

 

GENLINMIXED 

 /DATA_STRUCTURE SUBJECTS=participant_id*item_id 

 /FIELDS TARGET=ln_response_time 

 /TARGET_OPTIONS DISTRIBUTION=NORMAL LINK=IDENTITY 

 /FIXED EFFECTS=condition knowledge_context period 

                condition*knowledge_context sequence 

        USE_INTERCEPT=TRUE 

 /RANDOM USE_INTERCEPT=TRUE SUBJECTS=participant_id 

        COVARIANCE_TYPE=VARIANCE_COMPONENTS 

 /RANDOM USE_INTERCEPT=TRUE SUBJECTS=item_id 

        COVARIANCE_TYPE=VARIANCE_COMPONENTS 

 /BUILD_OPTIONS CONFIDENCE_LEVEL=95 

        DF_METHOD=SATTERTHWAITE COVB=MODEL. 

 

 

*********************************************************************** 

10. THEMATIC-PAIR SENSITIVITY MODELS 

***********************************************************************. 

 

* Accuracy. 

 

GENLINMIXED 

 /DATA_STRUCTURE SUBJECTS=participant_id*item_id 

 /FIELDS TARGET=correct 

 /TARGET_OPTIONS REFERENCE=0 DISTRIBUTION=BINOMIAL LINK=LOGIT 

 /FIXED EFFECTS=condition thematic_pair period sequence 

        USE_INTERCEPT=TRUE 

 /RANDOM USE_INTERCEPT=TRUE SUBJECTS=participant_id 

        COVARIANCE_TYPE=VARIANCE_COMPONENTS 

 /RANDOM USE_INTERCEPT=TRUE SUBJECTS=item_id 

        COVARIANCE_TYPE=VARIANCE_COMPONENTS 

 /BUILD_OPTIONS CONFIDENCE_LEVEL=95 

        DF_METHOD=SATTERTHWAITE COVB=MODEL. 

 

* Response expiration. 

 

GENLINMIXED 

 /DATA_STRUCTURE SUBJECTS=participant_id*item_id 

 /FIELDS TARGET=timeout 

 /TARGET_OPTIONS REFERENCE=0 DISTRIBUTION=BINOMIAL LINK=LOGIT 

 /FIXED EFFECTS=condition thematic_pair period sequence 

        USE_INTERCEPT=TRUE 



 /RANDOM USE_INTERCEPT=TRUE SUBJECTS=participant_id 

        COVARIANCE_TYPE=VARIANCE_COMPONENTS 

 /RANDOM USE_INTERCEPT=TRUE SUBJECTS=item_id 

        COVARIANCE_TYPE=VARIANCE_COMPONENTS 

 /BUILD_OPTIONS CONFIDENCE_LEVEL=95 

        DF_METHOD=SATTERTHWAITE COVB=MODEL 

 /EMMEANS TABLES=condition COMPARE=condition CONTRAST=PAIRWISE 

 /EMMEANS_OPTIONS SCALE=ORIGINAL PADJUST=LSD. 

 

* Response time. 

 

GENLINMIXED 

 /DATA_STRUCTURE SUBJECTS=participant_id*item_id 

 /FIELDS TARGET=ln_response_time 

 /TARGET_OPTIONS DISTRIBUTION=NORMAL LINK=IDENTITY 

 /FIXED EFFECTS=condition thematic_pair period sequence 

        USE_INTERCEPT=TRUE 

 /RANDOM USE_INTERCEPT=TRUE SUBJECTS=participant_id 

        COVARIANCE_TYPE=VARIANCE_COMPONENTS 

 /RANDOM USE_INTERCEPT=TRUE SUBJECTS=item_id 

        COVARIANCE_TYPE=VARIANCE_COMPONENTS 

 /BUILD_OPTIONS CONFIDENCE_LEVEL=95 

        DF_METHOD=SATTERTHWAITE COVB=MODEL. 

 

 

*********************************************************************** 

11. PARTICIPANT-LEVEL RANDOM-SLOPE ROBUSTNESS MODELS 

***********************************************************************. 

 

* Accuracy. 

 

GENLINMIXED 

 /DATA_STRUCTURE SUBJECTS=participant_id*item_id 

 /FIELDS TARGET=correct 

 /TARGET_OPTIONS REFERENCE=0 DISTRIBUTION=BINOMIAL LINK=LOGIT 

 /FIXED EFFECTS=condition knowledge_context period 

                condition*knowledge_context 

        USE_INTERCEPT=TRUE 

 /RANDOM EFFECTS=condition USE_INTERCEPT=TRUE 

        SUBJECTS=participant_id 

        COVARIANCE_TYPE=UNSTRUCTURED 

 /RANDOM USE_INTERCEPT=TRUE SUBJECTS=item_id 

        COVARIANCE_TYPE=VARIANCE_COMPONENTS 

 /BUILD_OPTIONS CONFIDENCE_LEVEL=95 

        DF_METHOD=SATTERTHWAITE COVB=MODEL. 

 

* Response expiration. 

* This specification produced the non-positive-definite random-effects 



* covariance matrix and Hessian reported in the manuscript. 

 

GENLINMIXED 

 /DATA_STRUCTURE SUBJECTS=participant_id*item_id 

 /FIELDS TARGET=timeout 

 /TARGET_OPTIONS REFERENCE=0 DISTRIBUTION=BINOMIAL LINK=LOGIT 

 /FIXED EFFECTS=condition knowledge_context period 

                condition*knowledge_context 

        USE_INTERCEPT=TRUE 

 /RANDOM EFFECTS=condition USE_INTERCEPT=TRUE 

        SUBJECTS=participant_id 

        COVARIANCE_TYPE=UNSTRUCTURED 

 /RANDOM USE_INTERCEPT=TRUE SUBJECTS=item_id 

        COVARIANCE_TYPE=VARIANCE_COMPONENTS 

 /BUILD_OPTIONS CONFIDENCE_LEVEL=95 

        DF_METHOD=SATTERTHWAITE COVB=MODEL. 

 

* Response time. 

 

GENLINMIXED 

 /DATA_STRUCTURE SUBJECTS=participant_id*item_id 

 /FIELDS TARGET=ln_response_time 

 /TARGET_OPTIONS DISTRIBUTION=NORMAL LINK=IDENTITY 

 /FIXED EFFECTS=condition knowledge_context period 

                condition*knowledge_context 

        USE_INTERCEPT=TRUE 

 /RANDOM EFFECTS=condition USE_INTERCEPT=TRUE 

        SUBJECTS=participant_id 

        COVARIANCE_TYPE=UNSTRUCTURED 

 /RANDOM USE_INTERCEPT=TRUE SUBJECTS=item_id 

        COVARIANCE_TYPE=VARIANCE_COMPONENTS 

 /BUILD_OPTIONS CONFIDENCE_LEVEL=95 

        DF_METHOD=SATTERTHWAITE COVB=MODEL. 

 

 

*********************************************************************** 

12. EXPLORATORY HOLM CORRECTION ACROSS THE THREE MAIN OUTCOMES 

***********************************************************************. 

 

* Unadjusted overall between-block p-values: 

* Response expiration = .009 

* Response time       = .492 

* Accuracy            = .909. 

* 

* Ordered Holm calculations: 

* .009 × 3 = .027 

* .492 × 2 = .984 

* .909 × 1 = .909, constrained to be no smaller than the preceding 



* adjusted value (.984). 

* 

* Final Holm-adjusted p-values: 

* Response expiration = .027 

* Response time        = .984 

* Accuracy             = .984. 

 

 

*********************************************************************** 

13. SYSTEM USABILITY SCALE 

***********************************************************************. 

 

* The following assumes the participant-level dataset contains the ten 

* individual SUS items as SUS1-SUS10 and the calculated total as SUS_total. 

* Replace these names if the analysis file uses different variable names. 

 

RELIABILITY 

 /VARIABLES=SUS1 SUS2 SUS3 SUS4 SUS5 SUS6 SUS7 SUS8 SUS9 SUS10 

 /SCALE('System Usability Scale') ALL 

 /MODEL=ALPHA. 

 

* Spearman correlations were calculated on one record per participant 

* between SUS total score and overall behavioral performance. 

 

NONPAR CORR 

 /VARIABLES=SUS_total overall_accuracy overall_response_time 

 /PRINT=SPEARMAN TWOTAIL 

 /MISSING=PAIRWISE. 


